
1) Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 11.2.4, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 5). Execution log files are provided as evidence. 

3) Verification Test Summary

Test Case:
TC_11_2_4
Test Group:
LTE/11/IMS_EmergencyCall.ttcn
ATS Version:
iwd-EUTRA-B2013-03_D13wk35

System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
INTEL XMM7160
Verification Status:
PASS
4) Corrections required for test case 11.2.4
Introduction

This section describes the changes required to make test 11.2.4 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2013-03_D13wk35 release.

4.1 Change 1

	Testcase name
	TC 11.2.4

	Reason for change
	1)  As procedure is different from 4.5A.5.3, corresponding code comment out, implemented step by step instead as below

2)   Acc. to 24.301 and section 5.4.2, authentication challenge is accepted only if USIM is present
Hence, authentication procedure needs to be skipped.

3)  Acc. to 24.301 and section 5.4.3.2, during an attach procedure for emergency services, security mode control procedure is initiated using NULL ciphering and integrity alogrithm and KSI is set to ‘000’.
Note: A draft prose CR would be raised for next RAN5 meeting for 2) and 3)
4)   Procedure 4.5A.5.3 of 36.508 from step 9-15 are added.
5)   As per change 7, only one PDN is required and receive the trigger from PDN1 after successful ecall.

6)   Receive trigger from PDN1 after ecall release



	Summary of change
	1. Commented the common fuction f_EUTRA_508_IMSEmergencyCallLimitedService_Steps5_15 
2. Autentication is skipped
3. Security procedure using NULL ciphering and integrity procedure
4. Remaining steps from 9-15 of procedure 4.5A.5.3 are added
5. Reception of trigger after successful ecall on PDN1
6. Reception of trigger after ecall release on PDN1

	Source of change
	LTE/11/IMS_EmergencyCall.ttcn

	MCC160 Comment
	1. Accepted in principle.
f_EUTRA_508_IMSEmergencyCallLimitedService_Steps5_15 improved for the additional needs of the present test case. 
2. Authentication skipped as envisaged in the generic procedure.
3. Security procedure using NULL ciphering and integrity procedure
4. Updated generic procedure used => steps 9-15 are included
5. Accepted in principle. See below.
6. Accepted


Before change:

	  function fl_TC_11_2_4_Body()runs on EUTRA_PTC

  {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var float v_T3412Value := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 360.0); // 6min

    timer t_WatchDog := 5.0;

    //The UE is switched on without a valid USIM inserted.

    //@siclog "Step 1" siclog@

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA); // @sic R5s120065 sic@

    //Causes the UE originate an emergency bearer service.

    //@siclog "Step 2" siclog@

    f_UT_RequestIMSCall (UT, px_EmergencyCallNumber);

    //Check: Does the UE transmits an RRCConnectionRequest message?

    //@siclog "Step 3" siclog@

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_emergency_Access);    // @sic R5-120721 change 3 sic@

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    //The SS transmits an RRCConnectionSetup message.

    //@siclog "Step 4 siclog@

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    // Check: Does the UE transmit an ATTACH REQUEST message to attach for emergency bearer services,

    // together with a PDN CONNECTIVITY REQUEST message for emergency bearer services?

    //@siclog "Step 5" siclog@

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,                                       // @sic R5s120121 MCC160 implementation sic@

                                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?),

                                                       cr_NAS_IndicationWithPiggybacking(?,

                                                                                          cdr_ATTACH_REQUEST_WithIMEI(tsc_EAttachType_Emergency,

                                                                                                                      v_AdditionalUpdateType),

                                                                                          cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_Emergency,

                                                                                                                          f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;

    // Steps 5 to 15 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 4.5A.5.3-1).

    //@siclog "Step 6-16" siclog@

    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps5_15(eutra_CellA,

                                                         v_NasInd,

                                                         NORMAL,

                                                         true);
     // EXCEPTION: (IP address allocation), steps 1 - 5 of annex C.22 should be performed in parallel

    //@siclog "Step 17" siclog@

    f_EUTRA_NAS_AttachComplete (eutra_CellA, tsc_EpsDefaultBearerId, dontWaitForIMS);

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST with

    // particular settings (use Reference dedicated EPS bearer context #1 - see table 6.6.2-1 in TS 36.508) (See Note 1).

    //@siclog "Step 18-19" siclog@

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 20" siclog@

    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps17_19(eutra_CellA);

    //IMS call has been established.

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN2]);

    //Release IMS Call

    //@siclog "Step 20A" siclog@

    f_UT_HangupCall(UT);

    //IMS call has been released.

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN2]);

    //The SS releases the RRC connection.

    //@siclog "Step 21" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    //The SS waits 6minutes. (Expire of T3412)

    //@siclog "Step 22" siclog@

    f_Delay (v_T3412Value);

    //@siclog "Step 23" siclog@

    t_WatchDog.start;

    alt {

        []SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellA, cr_RRCConnectionRequest(?))) {

          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 23");

        }

        [] t_WatchDog.timeout  {}

    }

  }// end fl_TC_11_2_4_Body


After change :

	function fl_TC_11_2_4_Body()runs on EUTRA_PTC

  {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NasInd;

    var NAS_UL_Message_Type v_PdnConnectivityRequest;

    var ProcedureTransactionIdentifier v_EPS_TI := '00'O;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var float v_T3412Value := f_EUTRA_SetTimerToleranceMax(eutra_CellA, nasTimer, 360.0); // 6min

    timer t_WatchDog := 5.0;


//1124 NK


template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v(tsc_GprsTimerUnit_6min, '00001'B);

    var NasCount_Type v_NasCountUL;

    var EUTRA_SecurityParams_Type v_AuthParams := f_EUTRA_Security_Get();

    var template (present) SecurityHeaderType v_ExpectedSecurityHeader;

    var GutiParameters_Type v_Guti_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_Guti_Params.PLMN_Identity);

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var template (value) AccessPointName v_APN;

    var template (value) ProtocolConfigOptions v_PcoToUE ;

    var template (omit) ProtocolConfigOptions v_PcoFromUE := omit;

    timer t_AuthTimer := 5.0;
    //The UE is switched on without a valid USIM inserted.

    //@siclog "Step 1" siclog@

    f_EUTRA_SwitchOnUEandStartIP(eutra_CellA); // @sic R5s120065 sic@

    //Causes the UE originate an emergency bearer service.

    //@siclog "Step 2" siclog@

    f_UT_RequestIMSCall (UT, px_EmergencyCallNumber);

    //Check: Does the UE transmits an RRCConnectionRequest message?

    //@siclog "Step 3" siclog@

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_emergency_Access);    // @sic R5-120721 change 3 sic@

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 3");

    //The SS transmits an RRCConnectionSetup message.

    //@siclog "Step 4 siclog@

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    // Check: Does the UE transmit an ATTACH REQUEST message to attach for emergency bearer services,

    // together with a PDN CONNECTIVITY REQUEST message for emergency bearer services?

    //@siclog "Step 5" siclog@

    v_NasInd := f_EUTRA_RRCConnectionSetupComplete_Def(eutra_CellA,                                       // @sic R5s120121 MCC160 implementation sic@

                                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?),

                                                       cr_NAS_IndicationWithPiggybacking(?,

                                                                                          cdr_ATTACH_REQUEST_WithIMEI(tsc_EAttachType_Emergency,

                                                                                                                      v_AdditionalUpdateType),

                                                                                          cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_Emergency,

                                                                                                                          f_GetPdnType())));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 5");

    v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;


/*


//1124 NK

    // Steps 5 to 15 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 4.5A.5.3-1).

    //@siclog "Step 6-16" siclog@

    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps5_15(eutra_CellA,

                                                         v_NasInd,

                                                         NORMAL,

                                                         true,















 p_T3412);

     // EXCEPTION: (IP address allocation), steps 1 - 5 of annex C.22 should be performed in parallel


   */
     //Steps 5 to 19 of the generic test procedure for IMS Emergency call establishment in EUTRA: Limited Service (TS 36.508 4.5A.5.3-1).

    //@siclog "Step 7-21" siclog@

    // Set UE network capabilities as sent by the UE to be used later (see TS 24.301 clause 5.4.3.3)

    v_AuthParams.NAS_SecurityCap := f_EUTRA_FillNAS_SecurityCapability(v_NasInd.Pdu.Msg.aTTACH_REQUEST);

    f_EUTRA_Security_Set(v_AuthParams);

    if (v_NasInd.SecurityProtection.Status == tsc_SHT_NoSecurityProtection) {

      v_ExpectedSecurityHeader := tsc_SHT_NoSecurityProtection;

    } else {

      v_ExpectedSecurityHeader := (tsc_SHT_NoSecurityProtection, tsc_SHT_IntegrityProtected);

    }

    v_PdnConnectivityRequest := v_NasInd.Pdu.PiggybackedPduList[0].Msg;

    v_EPS_TI := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.procedureTransactionIdentifier;

    // Authentication

    // Evaluate Authentication parameters and generate new keys

    v_AuthParams := f_EUTRA_Authentication_InitNAS(v_AuthParams, v_PLMN);

    //1124 NK


v_AuthParams.NAS_Integrity.Algorithm := '000'B; // EIA0

    v_AuthParams.NAS_Ciphering.Algorithm := '000'B; // EEA0

    v_AuthParams.AS_Integrity.Algorithm := eia0_v920;

    v_AuthParams.AS_Ciphering.Algorithm := eea0;

    v_AuthParams.KSIasme := '000'B;

    f_EUTRA_Security_Set(v_AuthParams);

    // NAS Security

    v_NasCountUL := f_EUTRA_NAS_ActivateSecurity(eutra_CellA, v_AuthParams);   // NAS SECURITY COMMAND & COMPLETE

    // ESM information optionally transferred;  @sic R5-110708 sic@

    v_APN := f_ESM_InitAPN (eutra_CellA, f_CheckEsmInfoTransferFlag(v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST), v_EPS_TI);

    // RRC Security Mode procedures

    v_AuthParams := f_EUTRA_RRC_ActivateSecurity(eutra_CellA, v_AuthParams, v_NasCountUL); // RRC SECURITY COMMAND & COMPLETE

    f_EUTRA_Security_Set(v_AuthParams); //Saving security parameters

    // UE Capability check

    f_EUTRA_Capability (eutra_CellA, tsc_RRC_TI_Def); // This enquires for the eutra capability but does not check the contents of the message received

    //1124  Network need to support emergecny calls to perform IMS emergency calls

    // Activate Def Bearer

    v_PcoFromUE := v_PdnConnectivityRequest.pDN_CONNECTIVITY_REQUEST.protocolConfigurationOptions;  // may be omit

    v_PcoToUE := f_GetDefaultProtocolConfigOptions (v_PcoFromUE); // @sic R5-104703 sic@

    f_EUTRA_IdleUpdated_Step14_15 (eutra_CellA,

                                   NORMAL,

                                   v_NasInd.Pdu.Msg.aTTACH_REQUEST.addUpdateType,

                                   v_EPS_TI,

                                   f_CheckPCOforIPallocationViaNas (v_PcoFromUE),

                                   v_PcoToUE,

                                   v_APN,

                                   -,p_T3412,-,-,-,-,

                                   cs_EPS_NwkFtSup_IMSVoice_EMCBS);

    //@siclog "Step 17" siclog@

    f_EUTRA_NAS_AttachComplete (eutra_CellA, tsc_EpsDefaultBearerId, dontWaitForIMS);
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    // The SS transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT REQUEST with

    // particular settings (use Reference dedicated EPS bearer context #1 - see table 6.6.2-1 in TS 36.508) (See Note 1).

    //@siclog "Step 18-19" siclog@

    // The UE transmits an ACTIVATE DEDICATED EPS BEARER CONTEXT ACCEPT message as specified.

    //@siclog "Step 20" siclog@

    f_EUTRA_508_IMSEmergencyCallLimitedService_Steps17_19(eutra_CellA);

    //IMS call has been established.

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);


//1124 NK

    //Release IMS Call

    //@siclog "Step 20A" siclog@

    f_UT_HangupCall(UT);

    //IMS call has been released.

    f_IMS_IPCAN_WaitForTrigger(IMS[tsc_Index_PDN1]);

    //The SS releases the RRC connection.

    //@siclog "Step 21" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    //The SS waits 6minutes. (Expire of T3412)

    //@siclog "Step 22" siclog@

    f_Delay (v_T3412Value);

    //@siclog "Step 23" siclog@

    t_WatchDog.start;

    alt {

        []SRB.receive(car_SRB0_RrcPdu_IND(eutra_CellA, cr_RRCConnectionRequest(?))) {

          f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 23");

        }

        [] t_WatchDog.timeout  {}

    }

  }// end fl_TC_11_2_4_Body


4.2 Change 2

	Testcase name
	TC 11.2.4

	Reason for change
	1) Invite request would be received with context A6 as per 34.229-1 and section A.2.1.
2) Release of ecall also happen in the context of A_2_1_A6

	Summary of change
	1) Passed the context as A_2_1_A6 as input to the Annex C22 procedure.
2) Passed the context as A_2_1_A6 as input to the Call release procedure

	Source of change
	LTE/11/IMS_EmergencyCall_IMS.ttcn

	MCC160 Comment
	Accepted in principle: 


Before change:

	function f_TC_11_2_4_IMS2() runs on IMS_PTC

  {

    var INVITE_Request v_InviteRequest;

    var boolean v_IsUnprotected := true;

    f_IMS_PTC_Init();

    // Emergency call setup: 34.229-1 Annex C.22

    v_InviteRequest := f_IMS_EmergencyCallSetup_AnnexC22(-, v_IsUnprotected);

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

    //Release Emergency Call MO

    f_IMS_CallReleaseMO(v_InviteRequest, -, -, v_IsUnprotected);

    // Inform EUTRA: IMS Call released.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

  }


After change :

	  function f_TC_11_2_4_IMS2() runs on IMS_PTC

  {

    var INVITE_Request v_InviteRequest;

    var boolean v_IsUnprotected := true;

    f_IMS_PTC_Init();


//1124 NK

    // Emergency call setup: 34.229-1 Annex C.22

    v_InviteRequest := f_IMS_EmergencyCallSetup_AnnexC22(A_2_1_A6, v_IsUnprotected);

    // Inform EUTRA: IMS Call established.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);


//1124 NK

    //Release Emergency Call MO

    f_IMS_CallReleaseMO(v_InviteRequest, -, -, v_IsUnprotected, A_2_1_A6);

    // Inform EUTRA: IMS Call released.

    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

  }


4.3 Change 3

	Testcase name
	TC 11.2.4

	Reason for change
	1) As per 24.301 Table 9.9.3.12.1, identity type has been created for IMEI

2) Created a new template which will take this new indentity type for IMEI

	Summary of change
	1) Declared new NAS_IdType for IMEI
2) Defined new template cr_MobileIdentityImei_2 to take new NAS_IdType for IMEI

	Source of change
	Common/NAS/NAS_CommonTemplates.ttcn

	MCC160 Comment
	Accepted in principle. See below.


	const NAS_IdType tsc_IdType_IMEI_2   := '011'B;

//1124 NK 

 template (present) MobileIdentity cr_MobileIdentityImei_2(B4_Type p_FirstDigit,

                                                          B1_Type p_OddEvenInd,

                                                          octetstring p_OtherDigits) :=

  {  /* @status    APPROVED (LTE, LTE_A) */

    iei         := omit,

    iel         := int2oct(lengthof(p_OtherDigits) + 1, 1),

    idDigit1    := p_FirstDigit,                        // 4 bits

    oddEvenInd  := p_OddEvenInd,                        // 1 bit

    typeOfId    := tsc_IdType_IMEI_2,                     // 3 bits

    otherDigits := p_OtherDigits

  };


MCC160 Implementation:

	const NAS_IdType tsc_IdType_IMEI_24301 := '011'B;

  template (present) MobileIdentity cr_MobileIdentityImei(B4_Type p_FirstDigit,

                                                          B1_Type p_OddEvenInd,

                                                          octetstring p_OtherDigits,












  NAS_IdType p_NAS_IdType := tsc_IdType_IMEI) := // @sic R5s130758 sic@

  {  /* @status    APPROVED (LTE, LTE_A) */

    iei         := omit,

    iel         := int2oct(lengthof(p_OtherDigits) + 1, 1),

    idDigit1    := p_FirstDigit,                        // 4 bits

    oddEvenInd  := p_OddEvenInd,                        // 1 bit

    typeOfId    := p_NAS_IdType,                        // 3 bits // @sic R5s130758 sic@
    otherDigits := p_OtherDigits

  };


4.4 Change 4

	Testcase name
	TC 11.2.4

	Reason for change
	As per change 3, the newly defined template is used from f_Imei2MobileIdentity_2 t o fill up IMEI id type

	Summary of change
	New function f_Imei2MobileIdentity_2 to fill up IMEI IE

	Source of change
	Common/NAS/NAS_AuxiliaryDefsAndFunctions.ttcn

	MCC160 Comment
	Accepted in principle. See below.


	//1124

  //----------------------------------------------------------------------------

  /*

   * @desc      Convert from IMEI of type hexstring to NAS MobileIdentity

   * @param     p_IMEI

   * @return    template (present) MobileIdentity

   * @status    APPROVED (LTE, LTE_A)

   */

  function f_Imei2MobileIdentity_2(hexstring p_IMEI) return template (present) MobileIdentity

  {

    var integer v_ImeiLength := lengthof(p_IMEI);

    var integer v_Odd := (v_ImeiLength rem 2);

    var octetstring v_Other  := fl_ImsiImei2Octetstring (p_IMEI);

    var B4_Type v_FirstDigit := hex2bit(p_IMEI[0]);

    var B1_Type v_OddEvenInd := int2bit(v_Odd, 1);

    return cr_MobileIdentityImei_2(v_FirstDigit, v_OddEvenInd, v_Other);

  }  


MCC160 Implementation:

	  function f_Imei2MobileIdentity(hexstring p_IMEI,









        NAS_IdType p_NAS_IdType := tsc_IdType_IMEI) return template (present) MobileIdentity // @sic R5s130758 sic@

  {

    var integer v_ImeiLength := lengthof(p_IMEI);

    var integer v_Odd := (v_ImeiLength rem 2);

    var octetstring v_Other  := fl_ImsiImei2Octetstring (p_IMEI);

    var B4_Type v_FirstDigit := hex2bit(p_IMEI[0]);

    var B1_Type v_OddEvenInd := int2bit(v_Odd, 1);

    return cr_MobileIdentityImei(v_FirstDigit, v_OddEvenInd, v_Other, tsc_IdType_IMEI); // @sic R5s130758 sic@

  }


4.5 Change 5
	Testcase name
	TC 11.2.4

	Reason for change
	As defined in Change 4, newly defined function is called from here so that ATTACH REQUEST will be filled up with right IE 

	Summary of change
	Update of ATTACH REQUEST for IME IE using the function defined under change 4

	Source of change
	LTE/11/IMS_EmergencyCall.ttcn

	MCC160 Comment
	Accepted in principle. See below.


Before change:

	  template (present) NAS_UL_Message_Type cdr_ATTACH_REQUEST_WithIMEI(template (present) NAS_AttDetValue_Type p_AttachTypeValue,

                                                                     template AdditionalUpdateType p_AdditionalUpdateType)

    modifies cr_508_ATTACH_REQUEST :=

  { /* 24.301 cl. 8.2.4 */

    /* @status    */

    aTTACH_REQUEST := {

      epsMobileIdentity   := f_Imei2MobileIdentity(px_IMEI_Def)

    }

  };

  


After change :

	  //1124 
  template (present) NAS_UL_Message_Type cdr_ATTACH_REQUEST_WithIMEI(template (present) NAS_AttDetValue_Type p_AttachTypeValue,

                                                                     template AdditionalUpdateType p_AdditionalUpdateType)

    modifies cr_508_ATTACH_REQUEST :=

  { /* 24.301 cl. 8.2.4 */

    /* @status    */

    aTTACH_REQUEST := {

      epsMobileIdentity   := f_Imei2MobileIdentity_2(px_IMEI_Def)

    }

  };


MCC160 Implementation:

	  template (present) NAS_UL_Message_Type cdr_ATTACH_REQUEST_WithIMEI(template (present) NAS_AttDetValue_Type p_AttachTypeValue,

                                                                     template AdditionalUpdateType p_AdditionalUpdateType)

    modifies cr_508_ATTACH_REQUEST :=

  { /* 24.301 cl. 8.2.4 */

    /* @status    */

    aTTACH_REQUEST := {

      epsMobileIdentity   := f_Imei2MobileIdentity(px_IMEI_Def, tsc_IdType_IMEI_24301)

    }

  };

  


4.6 Change 6

	Testcase name
	TC 11.2.4

	Reason for change
	1)  At step 22, as per prose, SS needs to wait for 6 minutes, hence, guard timer increased from 6 minutes to 12 minutes.
2)  As per change 7,  only one PDN is sufficient which is for emergency call handling

	Summary of change
	1) Guard timer value changed from 360 to 1200
2) Passed true as input to the so that only one PDN is created for emergency call

	Source of change
	LTE/source/LTE_EPS_TS_Testcases.ttcn

	MCC160 Comment
	Accepted in principle.
According to R5s130759 9 minutes (540 sec) are sufficient.


Before change:

	testcase TC_11_2_4() runs on MTC_LTE system SYSTEM_LTE {

    // @purpose

    //   Emergency bearer services / Normal cell / NO-IMSI / Attach / No EPS security context / PDN connect / Service request  / Timer T3412 expires

    var EUTRA_PTC    v_EUTRA      := null;

    var IMS_PTC      v_IMS1       := null;

    var IMS_PTC      v_IMS2       := null;

    timer t_GuardTimer := int2float(360);
    v_EUTRA := EUTRA_PTC.create alive;

    v_IMS2 := IMS_PTC.create alive;

    f_MTC_ConnectPTCs_LTE(v_EUTRA, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_11_2_4_EUTRA());

    v_IMS2.start(f_TC_11_2_4_IMS2());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }


After change :

	testcase TC_11_2_4() runs on MTC_LTE system SYSTEM_LTE {

    // @purpose

    //   Emergency bearer services / Normal cell / NO-IMSI / Attach / No EPS security context / PDN connect / Service request  / Timer T3412 expires

    var EUTRA_PTC    v_EUTRA      := null;

    var IMS_PTC      v_IMS1       := null;

    var IMS_PTC      v_IMS2       := null;

    timer t_GuardTimer := int2float(1200); //1124 
    v_EUTRA := EUTRA_PTC.create alive;

    v_IMS2 := IMS_PTC.create alive;

    f_MTC_ConnectPTCs_LTE(v_EUTRA, v_IMS1, v_IMS2,true);

    v_EUTRA.start(f_TC_11_2_4_EUTRA());

    v_IMS2.start(f_TC_11_2_4_IMS2());

    t_GuardTimer.start;

    f_MTC_MainLoop(t_GuardTimer);

  }


4.7 Change 7-18
Please refer to changes 4 to 15 of R5s130756 (verification of TS 36.523-1 test case 8.1.2.12).

Additional Comment from MCC160 on change 1 of R5s130748 is applied in this test case as well.
5) Execution Log Files
Intel UE

The Intel XMM7160 passed this test case on the Anritsu ME7832L LTE system in FDD band 1. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test Case Execution log file :

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
6 References

	[1]
	R5s130759: This archive comprises text format execution log file.

